

Constructive Computer Architecture

Sequential Circuits

Arvind


Computer Science & Artificial Intelligence Lab.
Massachusetts Institute of Technology

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-1

Combinational circuits



Such circuits have no cycles (feedback) or state elements


September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-2

Flip flop: The basic building block of Sequential Circuits

Edge-Triggered Flip-flop



Metastability





Data is sampled at the rising edge of the clock

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-3

Flip-flops with Write Enables




Data is captured only if EN is on

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-4

Registers



Register: A group of flip-flops with a common clock and enable

Register file: A group of registers with a common clock, input and output port(s)

September 14, 2016


<http://csg.csail.mit.edu/6.175>

L04-5

An example

Modulo-4 counter

Prev State	NextState	
q1q0	inc = 0	inc = 1
00	00	01
01	01	10
10	10	11
11	11	00



$$q_0^{t+1} = \sim \text{inc} \cdot q_0^t + \text{inc} \cdot \sim q_0^t$$


$$q_1^{t+1} = \sim \text{inc} \cdot q_1^t + \text{inc} \cdot \sim q_1^t \cdot q_0^t + \text{inc} \cdot q_1^t \cdot \sim q_0^t$$

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-6

Modulo-4 counter circuit



$$q_0^{t+1} = \sim inc \cdot q_0^t + inc \cdot \sim q_0^t$$

$$q_1^{t+1} = \sim inc \cdot q_1^t + inc \cdot \sim q_1^t \cdot q_0^t + inc \cdot q_1^t \cdot \sim q_0^t$$

"Optimized" logic

$$q_0^{t+1} = inc \oplus q_0^t$$

$$q_1^{t+1} = (inc == 1) ? q_0^t \oplus q_1^t : q_1^t$$

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-7

Finite State Machines (Sequential Ckts)

- ◆ A computer (in fact all digital hardware) is an FSM
- ◆ Neither State tables nor diagrams are suitable for describing very large digital designs
 - large circuits must be described in a modular fashion
 - as a collection of cooperating FSMs
- ◆ BSV is a modern programming language to describe cooperating FSMs
 - We will give various examples of FSMs in BSV

September 14, 2016


<http://csg.csail.mit.edu/6.175>

L04-8

modulo4 counter in BSV

```
module moduloCounter(Counter);
    Reg#(Bit#(2)) cnt <- mkReg(0);
    method Action inc;
        cnt <= (~cnt[1]&cnt[0] | cnt[1]&~cnt[0],
                  ~cnt[0]);
    endmethod
    method Bit#(2) read;
        return cnt;
    endmethod
endmodule
```

Can be
replaced by
 $cnt+1$



September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-9

Interface

- ◆ Modulo counter has the following interface, i.e., type

```
interface Counter;
    method Action inc;
    method Bit#(2) read;
endinterface
```

- ◆ An interface can have many different implementations

- For example, the numbers may be represented as Gray code

September 14, 2016


<http://csg.csail.mit.edu/6.175>

L04-10

FIFO Interface

```
interface Fifo#(numeric type size, type t);
    method Bool notFull;
    method Bool notEmpty;
    method Action enq(t x);
    method Action deq;
    method t first;
endinterface
```

- enq should be called only if notFull returns True;
- deq and first should be called only if notEmpty returns True



September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-11


One-Element FIFO Implementation

```
module mkCFFifo (Fifo#(1, t));
    Reg#(t) d <- mkRegU;
    Reg#(Bool) v <- mkReg(False);
    method Bool notFull;
        return !v;
    endmethod
    method Bool notEmpty;
        return v;
    endmethod
    method Action enq(t x);
        v <= True; d <= x;
    endmethod
    method Action deq;
        v <= False;
    endmethod
    method t first;
        return d;
    endmethod
endmodule
```

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-12



Two-Element FIFO

```

module mkCFFifo (Fifo#(2, t));
    Reg#(t) da <- mkRegU();
    Reg#(Bool) va <- mkReg(False);
    Reg#(t) db <- mkRegU();
    Reg#(Bool) vb <- mkReg(False);
    method Bool notFull; return !vb; endmethod
    method Bool notEmpty; return va; endmethod
    method Action enq(t x);
        if (va) begin db <= x; vb <= True; end
        else begin da <= x; va <= True; end
    endmethod
    method Action deq;
        if (vb) begin da <= db; vb <= False; end
        else begin va <= False; end
    endmethod
    method t first; return da; endmethod
endmodule

```

Assume, if there is only one element in the FIFO it resides in da

parallel composition of actions


no change in fifo interface

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-13

Switch



```

if (inQ.first.color == Red) begin
    redQ.enq(inQ.first.value); inQ.deq;
end else begin
    greenQ.enq(inQ.first.value); inQ.deq;
end

```

parallel composition of actions. Effect of inQ.deq is not visible to inQ.first

The code does not test for empty inQ or full redQ or full greenQ conditions!

```

let x = inQ.first;
if (x.color == Red) redQ.enq(x.value);
else greenQ.enq(x.value);
inQ.deq;

```


September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-14

Switch with empty/full tests on queues

1



```
if (inQ.notEmpty) begin
    if (inQ.first.color == Red) begin
        if (redQ.notFull) begin
            redQ.enq(inQ.first.value); inQ.deq;
        end
    end else begin
        if (greenQ.notFull) begin
            greenQ.enq(inQ.first.value); inQ.deq;
        end
    end
    inQ.deq; What's wrong if the deq is moved here?
end
```

Atomicity violation!


September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-15

Switch with counters

1



```
if (inQ.first.color == Red) begin
    redQ.enq(inQ.first.value); inQ.deq;
    redC <= redC+1;
end else begin
    greenQ.enq(inQ.first.value); inQ.deq;
    greenC <= greenC+1;
end
```

Ignoring full/empty conditions

September 14, 2016


<http://csg.csail.mit.edu/6.175>

L04-16

Shared counters

```
if (inQA.first.color == Red) begin  
    redQA.enq(inQA.first.value);  
    inQA.deq; redC <= redC+1;  
end else begin  
    greenQA.enq(inQA.first.value);  
    inQA.deq; greenC <= greenC+1;  
end;  
if (inQB.first.color == Red) begin  
    redQB.enq(inQB.first.value);  
    inQB.deq; redC <= redC+1;  
end else begin  
    greenQB.enq(inQB.first.value);  
    inQB.deq; greenC <= greenC+1;  
end
```

Ignoring full/empty conditions



What is wrong with this code?

Double write error

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-17

Double-write problem

- ◆ Parallel composition is illegal if a double-write possibility exists
- ◆ If the BSV compiler cannot prove that the predicates for writes into a register or a method call are mutually exclusive, it rejects the program

September 14, 2016

<http://csg.csail.mit.edu/6.175>

L04-18

Observations

- ◆ These programs are not very complex and yet it would have been tedious to express these programs in a state table or as a circuit directly
- ◆ BSV method calls are not available in Verilog/VHDL, and thus such programs sometimes require tedious programming
- ◆ Even the meaning of double-write errors is not standardized across tool implementations in Verilog